*Open Log file
log using "/Users/JTAFOYA1/My Drive/Research/Englishness_Americanness/P_UK13029248_BBC.log", replace

*Replication Files for "Divid Images: How the English Percieve Nationhood and How This Shapes Voting and Opinion"
*Eric Kaufmann, Birkbeck College, University of London, UK 
*David L. Leal, University of Texas at Austin, USA 
*Joe R. Tafoya, DePaul University, Chicago, USA 

*Part 1 of 2, focused on the UK

*Author of this code: Joe R. Tafoya, PhD (j.tafoya@depaul.edu)

*Platform: STATA


set more off

**Load Englishness survey
use "/Users/joetafoya/Google Drive/Research/Englishness_Americanness/P_UK13029248_BBC.dta", replace

*Natiional Frames: Types and Scale Constructions
{
*How strongly, if at all, do the following contribute to your English identity?
tab bbc6a // Countryside
tab bbc6b // Christian tradition
tab bbc6c // Diverse cultural life
tab bbc6d // History and heritage
tab bbc6e // Sports and leisure activitie
tab bbc6f // Poetry and literature and art
tab bbc6g // Dance and music

*How strongly, if at all, do you associate the following characteristics with English identity?
tab bbc7a // Sense of humor
tab bbc7b // Sense of fair play
tab bbc7c // Good manners
tab bbc7d // Tolerance
tab bbc7e // Welcoming
tab bbc7f // Friendliness
tab bbc7g // Generosity
tab bbc7h // Stiff upper lip
tab bbc7i // Outward looking
tab bbc7j // Loud
tab bbc7k // Liberal
tab bbc7l // Traditional
tab bbc7m // Yob culture
tab bbc7n // Plain-speaking

recode bbc6a bbc6b bbc6c bbc6d bbc6e bbc6f bbc6g bbc7a bbc7b bbc7c bbc7d ///
bbc7e bbc7f bbc7g bbc7h bbc7i bbc7j bbc7k bbc7l bbc7m bbc7n (1=4) (2=3) ///
(3=2) (4=1) (else=.), gen(country christian diverse history sports poetry ///
dance humor fair manners tolerance welcoming friendly generous stiff outward ///
loud liberal traditional yob plain)

*In general, which, if any, of the following do you think makes a person "English"?
tab bbc8a // Consider self to be English
tab bbc8b // Being born in England
tab bbc8c // Growing up in England
tab bbc8d // Speaking English
tab bbc8e // Speaking with an English accent
tab bbc8f // Having one English parent
tab bbc8g // Having two English parents
tab bbc8h // Having lived in England for up to 5 years
tab bbc8i // Having lived in England for 5 to 10 years
tab bbc8j // Having lived in England for over 10 years

factor country christian diverse history sports poetry dance humor fair ///
manners tolerance welcoming friendly generous stiff outward loud liberal ///
traditional yob plain

factor country christian diverse history sports poetry dance stiff loud ///
traditional plain // Loud & Plain below to a seperate factor

factor humor fair manners tolerance welcoming friendly generous outward ///
liberal yob

factor country christian diverse history sports poetry dance stiff traditional
factor humor fair manners tolerance generous liberal
factor welcoming friendly outward loud yob plain 

alpha country christian diverse history sports poetry dance stiff ///
traditional, gen (traditions) // .797 Alpha
lab var traditions "Traditional"
tab traditions

alpha humor fair manners tolerance generous liberal, gen(values)
// .825 Alpha 
lab var values "Value"
tab values

alpha welcoming friendly outward loud yob plain, gen(expressive)
*egen expressive= rowmean(welcoming friendly outward loud yob plain)
// .619 Alpha
lab var expressive "Expressive"
tab expressive
}

*Demographics
{
*Which party did you vote for in 2015?
tab pastvote_2015x
recode pastvote_2015x 1=1 2=0 3=0 4=. 5=0 6=1 7=0 8=1 else=., gen(vote2015)
lab var vote2015 "Voted Conservative '15"
tab vote2015

*Which party did you vote for in 2017?
tab pastvote_2017
recode pastvote_2017 1=1 2=0 3=0 4=. 5=0 6=1 7=0 else=., gen(right)
tab right

*Past vote on EU referendum
tab pastvote_EURef
recode pastvote_EURef 1=0 2=1 3/4=., gen(referendum)
tab referendum

*Political Attention
tab political_attention
tab political_attention_2017
rename political_attention_2017 attention
lab var attention "Political Attention"
tab attention

*Describe self on a political scale
tab politics_scale_profile_update
recode politics_scale_profile_update 1/3=1 4=2 5/7=3 8=., gen(ideo3)
lab def ideo3a 1"Left" 2"Centre" 3"Right"
lab var ideo3 "Ideology"
lab val ideo3 ideo3a
tab ideo3

recode ideo3 3=1 else=0, gen(conservative)
lab def conservative2 0"Non-Conservative" 1"Conservative"
lab var conservative "Conservative"
lab val conservative conservative2
tab conservative

*Age
tab profile_bpcage
rename profile_bpcage age
lab var age "Age"
tab age

*Education
tab highest_education_gce
tab highest_education_simple
rename highest_education_simple education
lab var education "Education"
tab education

*Ethnicity
tab profile_ethnicity
recode profile_ethnicity 1/2=1 3/16=0, gen(white)
lab var white "White"
lab def white1 0"Non-White" 1"White"
lab val white white1
tab white

*Gender
tab profile_gender
recode profile_gender 1=0 2=1, gen(female)
lab var female "Female"
tab female

*Household Income
tab xprofile_gross_household
recode xprofile_gross_household 16/17=., gen(income)
lab var income "Income"
tab income

*Religious/non-religious
tab profile_religion,m
recode profile_religion 1=0 else=1, gen(relig)
lab var relig "White"
lab def relig1 0"Non-Religious" 1"Religious"
lab val relig relig1
tab relig
}



*Figure 1: "UK: Frame Clusterings"
	graph bar (mean) values traditions expressive, title(Aggregate) ysc(r(1) ///
	titleg(1)) ylab(0(1)4) blabel(bar, position(center) format( %4.0g) size(3) ///
	color(white)) bargap(30) legend(label(1 "Value Frames") ///
	label(2 "Traditional Frames") label(3 "Expressive Frames") size(3) ///
	rowgap(*.45) pos(6) rows(1) region(style(legend) fcolor(gs15) ///
	margin(medsmall))) ytitle(support scale) 
	graph save cluster1.gph, replace

	*Frame Clusterings*Race
	graph bar (mean) values traditions expressive, over(white) title(Race) ///
	ysc(r(1) titleg(1)) ylab(0(1)4) blabel(bar, position(center) ///
	format( %4.0g) size(3) color(white)) bargap(-15) ///
	legend(label(1 "Value Frames") label(2 "Traditional Frames") ///
	label(3 "Expressive Frames") size(3) rowgap(*.45) pos(6) rows(1) ///
	region(style(legend) fcolor(gs15) margin(medsmall))) ytitle(support scale) 
	graph save cluster2.gph, replace	
	
	*Frame Clusterings*Ideology
	graph bar (mean) values traditions expressive, over(conservative) ///
	title(Ideology) ysc(r(1) titleg(1)) ylab(0(1)4) blabel(bar, position(center) ///
	format( %4.0g) size(3) color(white)) bargap(-15) ///
	legend(label(1 "Value Frames") label(2 "Traditional Frames") ///
	label(3 "Expressive Frames") size(3) rowgap(*.45) pos(6) rows(1) ///
	region(style(legend) fcolor(gs15) margin(medsmall))) ytitle(support scale) 
	graph save cluster3.gph, replace
	
	*Frame Clusterings*Religiosity
	graph bar (mean) values traditions expressive, over(relig) ///
	title(Religiosity) ysc(r(1) titleg(1)) ylab(0(1)4) blabel(bar, ///
	position(center) format( %4.0g) size(3) color(white)) bargap(-15) ///
	legend(label(1 "Value Frames") label(2 "Traditional Frames") ///
	label(3 "Expressive Frames") size(3) rowgap(*.45) pos(6) rows(1) ///
	region(style(legend) fcolor(gs15) margin(medsmall))) ytitle(support scale) 
	graph save cluster4.gph, replace	

	grc1leg cluster1.gph cluster2.gph cluster3.gph cluster4.gph, ///
	ycommon col(2) title("UK: Frame Clusterings") ///
	note("Factor 1 Eigenvalue: Tradition (2.92), Value (2.62), Expression (2.03)" "Scale Reliability: Tradition (.797), Value (.825), Expression (.619)", size(3) pos(6))



*Figure 2: "UK Value Frames"
	graph bar (mean) humor manners fair tolerance generous liberal, ///
	title(Aggregate) ysc(r(1) titleg(1)) ylab(0(1)4) blabel(bar, ///
	position(center) format( %4.0g) size(3) color(white)) bargap(30) ///
	legend(label(1 "Sense of humor") label(2 "Good manners") ///
	label(3 "Sense of fair play") label(4 "Tolerance") label(5 "Generosity") ///
	label(6 "Liberal") size(2.5) rowgap(*.45) pos(6) rows(2) ///
	region(style(legend) fcolor(gs15) margin(medsmall))) ///
	ytitle(support scale)
	graph save value1.gph, replace

	*Values*Race
	graph bar (mean) humor manners fair tolerance generous liberal, ///
	over(white) title(Race) ysc(r(1) titleg(1)) ylab(0(1)4) blabel(bar, ///
	position(center) format( %4.0g) size(2) color(white)) bargap(-15) ///
	legend(label(1 "Sense of humor") label(2 "Good manners") ///
	label(3 "Sense of fair play") label(4 "Tolerance") label(5 "Generosity") ///
	label(6 "Liberal") size(2.5) rowgap(*.45) pos(6) rows(2) ///
	region(style(legend) fcolor(gs15) margin(medsmall))) ///
	ytitle(support scale)
	graph save value2.gph, replace	
	
	*Values*Ideology
	graph bar (mean) humor manners fair tolerance generous liberal, ///
	over(conservative) title(Ideology) ysc(r(1) titleg(1)) ylab(0(1)4) blabel(bar, ///
	position(center) format( %4.0g) size(2) color(white)) bargap(-15) ///
	legend(label(1 "Sense of humor") label(2 "Good manners") ///
	label(3 "Sense of fair play") label(4 "Tolerance") label(5 "Generosity") ///
	label(6 "Liberal") size(2.5) rowgap(*.45) pos(6) rows(2) ///
	region(style(legend) fcolor(gs15) margin(medsmall))) ///
	ytitle(support scale)
	graph save value3.gph, replace	
	
	*Values*Religiosity
	graph bar (mean) humor manners fair tolerance generous liberal, ///
	over(relig) title(Religiosity) ysc(r(1) titleg(1)) ylab(0(1)4) blabel(bar, ///
	position(center) format( %4.0g) size(2) color(white)) bargap(-15) ///
	legend(label(1 "Sense of humor") label(2 "Good manners") ///
	label(3 "Sense of fair play") label(4 "Tolerance") label(5 "Generosity") ///
	label(6 "Liberal") size(2.5) rowgap(*.45) pos(6) rows(2) ///
	region(style(legend) fcolor(gs15) margin(medsmall))) ///
	ytitle(support scale)	
	graph save value4.gph, replace	
	
	grc1leg value1.gph value2.gph value3.gph value4.gph, ycommon col(2) ///
	title("UK: Value Frames") note("Scale: Not at all (1), Not very strongly (2), Fairly strongly (3), Very strongly (4). Source: YouGov/BBC, March 2018", size(2) pos(6))

	
	
*Figure 3: "UK: Traditional Frames"
	graph bar (mean) country history traditional stiff poetry diverse sports ///
	dance christian, title(Aggregate) ysc(r(1) titleg(1)) ///
	ylab(0(1)4) blabel(bar, position(center) format( %4.0g) size(3) ///
	color(white)) bargap(30) legend(label(1 "Countryside") ///
	label(2 "History, heritage") label(3 "Traditional") ///
	label(4 "Stiff upper lip") label(5 "Poetry, literature, art") ///
	label(6 "Diverse cultural life") label(7 "Sports, leisure activities") ///
	label(8 "Dance, music") label(8 "Christian tradition") size(2.5) ///
	rowgap(*.45) pos(6) rows(3) region(style(legend) fcolor(gs15) ///
	margin(medsmall))) ytitle(support scale)
	graph save tradition1.gph, replace

	*Traditional*Race
	graph bar (mean) country history traditional stiff poetry diverse sports ///
	dance christian, over(white) title(Race) ysc(r(1) titleg(1)) ///
	ylab(0(1)4) blabel(bar, position(center) format( %4.0g) size(2) ///
	color(white)) bargap(-15) legend(label(1 "Countryside") ///
	label(2 "History, heritage") label(3 "Traditional") ///
	label(4 "Stiff upper lip") label(5 "Poetry, literature, art") ///
	label(6 "Diverse cultural life") label(7 "Sports, leisure activities") ///
	label(8 "Dance, music") label(8 "Christian tradition") size(2.5) ///
	rowgap(*.45) pos(6) rows(3) region(style(legend) fcolor(gs15) ///
	margin(medsmall))) ytitle(support scale)
	graph save tradition2.gph, replace
	
	*Traditional*Ideology
	graph bar (mean) country history traditional stiff poetry diverse sports ///
	dance christian, over(conservative) title(Ideology) ysc(r(1) titleg(1)) ///
	ylab(0(1)4) blabel(bar, position(center) format( %4.0g) size(2) ///
	color(white)) bargap(-15) legend(label(1 "Countryside") ///
	label(2 "History, heritage") label(3 "Traditional") ///
	label(4 "Stiff upper lip") label(5 "Poetry, literature, art") ///
	label(6 "Diverse cultural life") label(7 "Sports, leisure activities") ///
	label(8 "Dance, music") label(8 "Christian tradition") size(2.5) ///
	rowgap(*.45) pos(6) rows(3) region(style(legend) fcolor(gs15) ///
	margin(medsmall))) ytitle(support scale)
	graph save tradition3.gph, replace
	
	*Traditional*Religiosity
	graph bar (mean) country history traditional stiff poetry diverse sports ///
	dance christian, over(relig) title(Religiosity) ysc(r(1) titleg(1)) ///
	ylab(0(1)4) blabel(bar, position(center) format( %4.0g) size(2) ///
	color(white)) bargap(-15) legend(label(1 "Countryside") ///
	label(2 "History, heritage") label(3 "Traditional") ///
	label(4 "Stiff upper lip") label(5 "Poetry, literature, art") ///
	label(6 "Diverse cultural life") label(7 "Sports, leisure activities") ///
	label(8 "Dance, music") label(8 "Christian tradition") size(2.5) ///
	rowgap(*.45) pos(6) rows(3) region(style(legend) fcolor(gs15) ///
	margin(medsmall))) ytitle(support scale)
	graph save tradition4.gph, replace
	
	grc1leg tradition1.gph tradition2.gph tradition3.gph tradition4.gph, ///
	ycommon col(2) title("UK: Traditional Frames") ///
	note("Scale: Not at all (1), Not very strongly (2), Fairly strongly (3), Very strongly (4). Source: YouGov/BBC, March 2018", size(2) pos(6))

	
	
*Figure 4: "UK: Expressive Frames"
	graph bar (mean) friendly welcoming yob outward plain loud, title(Aggregate) ///
	ysc(r(1) titleg(1)) ylab(0(1)4) blabel(bar,position(center) format( %4.0g) ///
	size(3) color(white)) bargap(30) legend(label(1 "Friendliness") ///
	label(2 "Welcoming") label(3 "Yob culture") label(4 "Outward looking") ///
	label(5 "Plain-speaking") label(6 "Loud") size(2.5) rowgap(*.45) pos(6) ///
	rows(2) region(style(legend) fcolor(gs15) margin(medsmall))) ///
	ytitle(support scale)
	graph save expressive1.gph, replace

	*Expressive*Race
	graph bar (mean) friendly welcoming yob outward plain loud, over(white) ///
	title(Race) ysc(r(1) titleg(1)) ylab(0(1)4) blabel(bar,position(center) ///
	format( %4.0g) size(2) color(white)) bargap(-15) ///
	legend(label(1 "Friendliness") label(2 "Welcoming") ///
	label(3 "Yob culture") label(4 "Outward looking") ///
	label(5 "Plain-speaking") label(6 "Loud") size(2.5) rowgap(*.45) pos(6) ///
	rows(2) region(style(legend) fcolor(gs15) margin(medsmall))) ///
	ytitle(support scale)
	graph save expressive2.gph, replace
	
	*Expressive*Ideology
	graph bar (mean) friendly welcoming yob outward plain loud, ///
	over(conservative) title(Ideology) ysc(r(1) titleg(1)) ylab(0(1)4) ///
	blabel(bar,position(center) format( %4.0g) size(2) color(white)) ///
	bargap(-15) legend(label(1 "Friendliness") label(2 "Welcoming") ///
	label(3 "Yob culture") label(4 "Outward looking") ///
	label(5 "Plain-speaking") label(6 "Loud") size(2.5) rowgap(*.45) pos(6) ///
	rows(2) region(style(legend) fcolor(gs15) margin(medsmall))) ///
	ytitle(support scale)
	graph save expressive3.gph, replace	
	
	*Expressive*Religiosity
	graph bar (mean) friendly welcoming yob outward plain loud, ///
	over(relig) title(Religiosity) ysc(r(1) titleg(1)) ylab(0(1)4) ///
	blabel(bar,position(center) format( %4.0g) size(2) color(white)) ///
	bargap(-15) legend(label(1 "Friendliness") label(2 "Welcoming") ///
	label(3 "Yob culture") label(4 "Outward looking") ///
	label(5 "Plain-speaking") label(6 "Loud") size(2.5) rowgap(*.45) pos(6) ///
	rows(2) region(style(legend) fcolor(gs15) margin(medsmall))) ///
	ytitle(support scale)
	graph save expressive4.gph, replace		

	grc1leg expressive1.gph expressive2.gph expressive3.gph expressive4.gph, ///
	ycommon col(2) title("UK: Expressive Frames") ///
	note("Scale: Not at all (1), Not very strongly (2), Fairly strongly (3), Very strongly (4). Source: YouGov/BBC, March 2018", size(2) pos(6))

	
	
*Figure 5: "UK: Predictors of Frame Clusterings"	
	reg values age education income female attention i.conservative i.white i.relig
	eststo a
	reg traditions age education income female attention i.conservative i.white ///
	i.relig
	eststo b
	reg expressive age education income female attention i.conservative i.white ///
	i.relig
	eststo c

	coefplot (a, label(Value Frames)) (b, label(Traditional Frames)) ///
	(c, label(Expressive Frames)), drop(_cons) xline(0) bycoefs legend(size(3)  ///
	rowgap(*.45) pos(6) rows(1) region(style(legend) fcolor(gs15) margin(medsmall)))

	
	
*Figure 6: "UK: Vote Choice"
	logit referendum values traditions expressive age education income female ///
	attention i.conservative i.white i.relig
	eststo referendum1	

	logit right values traditions expressive age education income female ///
	attention i.conservative i.white i.relig, or
	eststo right1

	coefplot (right1, label("Voted Right")) (referendum1, ///
	label("Voted Brexit")), xline(0) drop(_cons) ///
	title("UK: Vote Choice") subtitle("Logistic Regression Coefficients, 95% CIs") ///
	ysize(8) nolabels byopts(legend(on)) ///
	coefl(values="Values" traditions="Traditional" ///
	expressive="Expressive" age="Age" education="Education" income="Income" ///
	female="Female" attention="Political Attention" ///
	1.conservative="Conservative" 1.white="White" 1.relig="Religious")
	
	
	
*Figure 7: "UK: Effect of National Frames on Conservative Vote '17"
	logit right values traditions expressive age education income ///
	female attention i.b2.ideo3 white, or

	margins, at(values=(0(1)4))
	marginsplot, title("Value Frames") ytitle("Pr(Voted)") ///
	xtitle("Support scale") ylabel(0(.20)1) note("Positive significant effect", pos(6))
	graph save 1.gph, replace

	margins, at(traditions=(0(1)4))
	marginsplot, title("Traditional Frames") ytitle("Pr(Voted)") ///
	xtitle("Support scale") ylabel(0(.20)1) note("Positive significant effect", pos(6))
	graph save 2.gph, replace

	margins, at(expressive=(0(1)4))
	marginsplot, title("Expression Frames") ytitle("Pr(Voted)") ///
	xtitle("Support scale") ylabel(0(.20)1) note("Positive significant effect", pos(6))
	graph save 3.gph, replace

	grc1leg 1.gph 2.gph 3.gph, ycommon col(3) ///
	title("UK: Effect of National Frames on Conservative Vote '17") ///
	subtitle("Marginal Predictions w/95% CIs")	

	
	
	
*Figure 8: "UK: Effect of National Frames on Leave EU Vote"
	logit referendum values traditions expressive age education income ///
	female attention i.b2.ideo3 white, or
	eststo referendum

	margins, at(values=(0(1)4))
	marginsplot, title("Value Frames") ytitle("Pr(Voted)") ///
	xtitle("Support scale") ylabel(0(.20)1) note("Positive significant effect", pos(6))
	graph save 1.gph, replace

	margins, at(traditions=(0(1)4))
	marginsplot, title("Traditional Frames") ytitle("Pr(Voted)") ///
	xtitle("Support scale") ylabel(0(.20)1) note("Positive significant effect", pos(6))
	graph save 2.gph, replace

	margins, at(expressive=(0(1)4))
	marginsplot, title("Expression Frames") ytitle("Pr(Voted)") ///
	xtitle("Support scale") ylabel(0(.20)1) note("Positive significant effect", pos(6))
	graph save 3.gph, replace

	grc1leg 1.gph 2.gph 3.gph, ycommon col(3) ///
	title("UK: Effect of National Frames on Leave EU Vote") ///
	subtitle("Marginal Predictions w/95% CIs")	

	
	
	
*Figure 9: "Marginal Effects of Value Frames on Referendum Vote"		
	logit referendum i.conservative##c.values i.conservative##c.traditions ///
	i.conservative##c.expressive age education income female attention ///
	i.white i.relig
	eststo referendum3
	
		margins conservative, dydx(values expressive) predict(pr)
		
		margins conservative, at(values=(1(.5)4)) predict(pr)
		marginsplot, xtitle("Values Frames") ///
		ylab(0(.20)1) ytitle("Pr(Voting for Brexit)") yline(.50, lpattern(dot)) ///
		title("Marginal Effects of Values Frames on Referendum Vote") ///
		subtitle("(by Group ID w/95 CIs)")
		graph save referendum1.gph, replace		

		
		
		
*Figure 10: "Marginal Effect of Expressive Frames on Voting in Elections"
	logit right i.conservative##c.values i.conservative##c.traditions ///
	i.conservative##c.expressive age education income female attention ///
	i.white i.relig
	eststo right3
		
		margins conservative, dydx(expressive) predict(pr)
		margins conservative, at(expressive=(1(.5)4)) predict(pr)
		marginsplot, title("Voted Right") xtitle("Expressive Frames") ///
		ylab(0(.20)1) ytitle("Pr(Voting for Right Party)") yline(.50, lpattern(dot))
		graph save right2.gph, replace	
			
	logit referendum i.conservative##c.values i.conservative##c.traditions ///
	i.conservative##c.expressive age education income female attention ///
	i.white i.relig
	eststo referendum3		
	
		margins conservative, at(expressive=(1(.5)4)) predict(pr)
		marginsplot, title("Voted Brexit") xtitle("Expressive Frames") ///
		ylab(0(.20)1) ytitle("Pr(Voting for Brexit)") yline(.50, lpattern(dot))
		graph save referendum2.gph, replace	
		
	grc1leg right2.gph referendum2.gph, ycommon col(2) ///
	title("Marginal Effects of Expressive Frames on Voting in Elections") ///
	subtitle("(by Group ID w/95 CIs)")


	
	
*Figure 11: "Marginal Effects of Expressive Frames on Voting in Elections"
	logit right i.relig##c.values i.relig##c.traditions ///
	i.relig##c.expressive age education income female attention ///
	i.conservative i.white
	eststo right4
	
		margins relig, dydx(expressive) predict(pr)
		margins relig, at(expressive=(1(.5)4)) predict(pr)
		marginsplot, title("Voted Right") xtitle("Expressive Frames") ///
		ylab(0(.20)1) ytitle("Pr(Voting for Right Party)") yline(.50, lpattern(dot))
		graph save right3.gph, replace		
	
	logit referendum i.relig##c.values i.relig##c.traditions ///
	i.relig##c.expressive age education income female attention ///
	i.conservative i.white
	eststo referendum4
	
		margins relig, dydx(expressive) predict(pr)
		margins relig, at(expressive=(1(.5)4)) predict(pr)
		marginsplot, title("Voted Brexit") xtitle("Expressive Frames") ///
		ylab(0(.20)1) ytitle("Pr(Voting for Brexit)") yline(.50, lpattern(dot))
		graph save referendum4.gph, replace	
		
		grc1leg right3.gph referendum4.gph, ycommon col(2) ///
		title("Marginal Effects of Expressive Frames on Voting in Elections") ///
		subtitle("(by Group ID w/95 CIs)")		
	


